

Appendix

Textual Notation Syntax Definition

This appendix defines the syntax of the textual notation used in this thesis to represent behaviour. The definition is expressed in Backus-Naur Form.

A.1 Behaviour definition

```
behaviour-definition = behaviour-identifier ":@" definition-block.
definition-block = "{" [synchronization-requirements-list] causality-relation
  {"", " causality-relation"} [local-definitions] "}".
local-definitions = "where" behaviour-definition {behaviour-definition}.
```

A.2 Synchronization requirements

```
synchronization-requirements-list =
  "(" synchronization-requirement {";" synchronization-requirement} ")".
synchronization-requirement =
  behaviour-identifier "," behaviour-identifier {"", " behaviour-identifier"} ":"
  (action-identifier | interaction-identifier) {"", " (action-identifier | interaction-identifier)}.
```

A.3 Causality relation

```
causality-relation = condition "→" result.
condition = simple-condition | composite-condition.
simple-condition =
  start | entry-point | occurrence-condition | non-occurrence-condition | behaviour-exit.
start = "start" [attributes-list [constraints-list]].
entry-point = entry-identifier [attributes-list [constraints-list]].
occurrence-condition = enabling-action | enabling-interaction-contribution.
non-occurrence-condition =
  ("¬" disabling-action) | ("¬" disabling-interaction-contribution).
behaviour-exit = behaviour-identifier "(" exit-point {"", " exit-point} ")".
```

result = result-action | result-interaction-contribution | exit-point | behaviour-entry.
exit-point = exit-identifier [attributes-list [constraints-list]].
composite-condition =
 (simple-condition ("^" | "v") (simple-condition | composite-condition)) |
 "(" (simple-condition ("^" | "v") (simple-condition | composite-condition)) ")".

A.4 Action and interaction

enabling-action = action.
disabling-action = action.
result-action = action.
action = action-identifier [attributes-list [constraints-list]].
enabling-interaction-contribution = interaction-contribution.
disabling-interaction-contribution = interaction-contribution.
result-interaction-contribution = interaction-contribution.
interaction-contribution = interaction-identifier [attributes-list [constraints-list]].

A.5 Attributes

attributes-list = "(" location-time-information-list | (")" functionality-attribute) |
 (location-time-information-list "/" functionality-attribute) ")".
location-time-information-list =
 (location-attribute ["," time-information-list]) | time-information-list.
time-information-list = (time-attribute ["," information-attribute]) | information-attribute.
location-attribute = location-value-identifier ":" "*Location*".
time-attribute = time-value-identifier ":" "*Time*".
information-attribute = information-element {"," information-element}.
information-element = information-element-value-identifier ":"
 information-type-identifier.
functionality-attribute = extended-location-time-information-list.
extended-location-time-information-list =
 (extended-location-attribute ["," extended-time-information-list]) |
 extended-time-information-list.
extended-time-information-list =
 (extended-time-attribute ["," information-attribute]) | information-attribute.
extended-location-attribute = location-attribute {"," location-attribute}.
extended-time-attribute = time-attribute {"," time-attribute}.

A.6 Constraints

constraints-list = "/" constraint {"," constraint} "/".
constraint = simple-constraint | composite-constraint.

simple-constraint =
 (attribute-value-identifier "=" term-expression) |
 (term-expression operation-identifier term-expression).
 term-expression = simple-term | composite-term.
 simple-term = attribute-value-identifier | attribute-value.
 composite-term =
 (simple-term operation-identifier (simple-term | composite-term)) |
 "(" (simple-term operation-identifier (simple-term | composite-term)) ")".
 composite-constraint =
 (simple-constraint ("^" | "v") (simple-constraint | composite-constraint)) |
 "(" (simple-constraint ("^" | "v") (simple-constraint | composite-constraint)) ")".

A.7 Identifiers

behaviour-identifier = identifier.
 entry-identifier = "entry" {digit}.
 exit-identifier = "exit" {digit}.
 action-identifier = identifier.
 interaction-contribution-identifier = underlined-identifier.
 attribute-value-identifier =
 location-value-identifier | time-value-identifier | information-element-value-identifier.
 location-value-identifier = identifier.
 time-value-identifier = identifier.
 information-element-value-identifier = identifier.
 information-type-identifier = identifier.
 attribute-value = identifier.
 operation-identifier = identifier | special-character.
 identifier = letter [{normal-character}].
 underlined-identifier = underlined-letter [{underlined-normal-character}].

A.8 Characters

digit = "0" | "1" | ... | "9".
 underlined-digit = "0" | "1" | ... | "9".
 letter = "a" | "b" | ... | "z" | "A" | "B" | ... | "Z".
 underlined-letter = "a" | "b" | ... | "z" | "A" | "B" | ... | "Z".
 normal-character = letter | digit.
 underlined-normal-character = underlined-letter | underlined-digit.
 special-character = "+" | "-" | "x" | "÷" | "=" | "<" | ">" | "≤" | "≥".

